On Enumeration of Hypergroups of Order 3

Ch. Tsitouras, [1] Ch. G. Massouros [2]

TEI of Chalkis, Department of Applied Sciences, GR34400 Psahna, Greece

Abstract

In this paper we present a symbolic manipulation package that enumerates the hypergroups of order 3. It separates them to isomorphic classes and calculates their cardinality.

Key words: Hypercomposition, Isomorphism.
1991 MSC: 20N20, 68W30

1 Introduction

1934 was the year that Frederic Marty defined the hypergroup [1] and so the time that the theory of hyper–compositional structures was born. Over the years hyper–compositional structures used in algebra, geometry, convexity, automata theory and even in some applied sciences. For the self-sufficiency of this paper, some definitions are recalled. Thus a *partial hypergroupoid* is a the pair $(H, \cdot)$, where H is a nonempty set and "$\cdot$" is a *hypercomposition* in H, i.e. a function from $H \times H$ to the powerset $P(H)$ of H. If the map is from $H \times H$ to the family of the non empty subsets of H, then $(H, \cdot)$ is called hypergroupoid. The axioms which endow $(H, \cdot)$ with the *hypergroup* structure are:

i. $a(bc) = (ab)c$ for every $a, b, c \in H$ (associativity)

ii. $aH = Ha = H$ for every $a \in H$ (reproductivity)

In a hypergroup, the result of the hypercomposition is always a nonempty set. Indeed, let $ab = \emptyset$, then $H = aH = a(bH) = (ab)H = \emptyset = \emptyset$ which is absurd, Mittas [5]. If only (i) is valid then $(H, \cdot)$ is called semihypergroup, while it is called quasi-hypergroup if only (ii) holds.

[1] E-Mail: tsitoura@teihal.gr, URL address: http://users.ntua.gr/tsitoura/
[2] E-mail: masouros@teihal.gr

Extend the hypercomposition "$\cdot$" from H to $P(H)$, by setting for all $A, B \in P(H)$

$$A \cdot B = \bigcup_{(a,b)\in A\times B} a \cdot b$$

$A \cdot b$ and $a \cdot B$ will have the same meaning as $A \cdot \{b\}$ and $\{a\} \cdot B$ respectively. Also, when nothing opposes it, there is no distinction between the elements and their corresponding singletons.

Hypergroups are much more flexible and varied than groups. For example if H is of prime cardinality p, there is a large number of non isomorphic hypergroups on H, while, up to isomorphic, only one group Z_p. This becomes clear in this paper, which enumerates the hypergroups of cardinality 3.

2 The method

A hypergroupoid is a set $H \neq \emptyset$ with a hypercomposition "$\cdot$" which is not necessarily associative or reproductive. Regarding the notification of the elements of the hypergroupoids of order 3, it can be assumed that they share the set $H = \{1, 2, 3\}$.

The hypercompositions in H are defined through the following table

$$\begin{array}{cccc} \cdot & 1 & 2 & 3 \\ 1 & a_{11} & a_{12} & a_{13} \\ 2 & a_{21} & a_{22} & a_{23} \\ 3 & a_{31} & a_{32} & a_{33} \end{array} \tag{1}$$

where $a_{ij} \subseteq H, i, j = 1, 2, 3$. The elements a_{ij}'s are chosen among the seven element set

$$\Lambda = P(H)\backslash\emptyset = \{\{1\}, \{2\}, \{3\}, \{1, 2\}, \{1, 3\}, \{2, 3\}, \{1, 2, 3\}\}.$$

If $\mathcal{H}_3$ denotes the set of all hypergroupoids of third order then for its cardinality holds: $|\mathcal{H}_3| = 7^9 = 40353607$. Migliorato [4] found, by computer, the total number $N_3 = 23192$ of hypergroups of order 3 while Nordo [6] computed using a program written in PASCAL the number $S_3 = 3999$ of non isomorphic hypergroups of the same order.

The mathematica package given in the Appendix is based on two functions, namely `ReproductivityTest[ ]` and `AssociativityTest[ ]`, which check out the corresponding properties. Their argument is a set of hypergroupoids in a list and their output is a True/False table. This package checks if hypergroupoids of any order form a hypergroup.

The reproductivity of the hypercompositions defined in (1) can be checked, through the verification of validity of the equivalent (to this axiom) equalities:

$$\bigcup_{j=1}^{3} a_{ij} = H, \text{ for } i = 1,2,3 \quad \text{and} \quad \bigcup_{i=1}^{3} a_{ij} = H, \text{ for } j = 1,2,3. \tag{2}$$

The cases that pass successfully this first test (i.e. the reproductivity's validity test) are going through the associativity's validity test, which is checking all the 27 possible triples $a(bc) = (ab)c$.

2.1 Classes of isomorphism

A hypergroupoid of order 3, is isomorphic with another 5 hypergroupoids. This derives from interchanges among the elements of the set H. More precisely

(i)	keep 1	interchange 2,3	
(ii)	keep 3	interchange 1,2	
(iii)	change 1 by 2	change 2 by 3	change 3 by 1
(iv)	change 1 by 3	change 2 by 1	change 3 by 2
(v)	keep 2	interchange 1,3	

So for the above matrix of hypercomposition (1) there derive the following five isomorphic hypercompositions:

(i)	1	2	3	(ii)	1	2	3	(iii)	1	2	3	(iv)	1	2	3	(v)	1	2	3
1	$\tilde{a}_{11}$	$\tilde{a}_{13}$	$\tilde{a}_{12}$		$\tilde{a}_{22}$	$\tilde{a}_{21}$	$\tilde{a}_{23}$		$\tilde{a}_{33}$	$\tilde{a}_{31}$	$\tilde{a}_{32}$		$\tilde{a}_{22}$	$\tilde{a}_{23}$	$\tilde{a}_{21}$		$\tilde{a}_{33}$	$\tilde{a}_{32}$	$\tilde{a}_{31}$
2	$\tilde{a}_{31}$	$\tilde{a}_{33}$	$\tilde{a}_{32}$		$\tilde{a}_{12}$	$\tilde{a}_{11}$	$\tilde{a}_{13}$		$\tilde{a}_{13}$	$\tilde{a}_{11}$	$\tilde{a}_{12}$		$\tilde{a}_{32}$	$\tilde{a}_{33}$	$\tilde{a}_{31}$		$\tilde{a}_{23}$	$\tilde{a}_{22}$	$\tilde{a}_{21}$
3	$\tilde{a}_{21}$	$\tilde{a}_{23}$	$\tilde{a}_{22}$		$\tilde{a}_{32}$	$\tilde{a}_{31}$	$\tilde{a}_{33}$		$\tilde{a}_{23}$	$\tilde{a}_{21}$	$\tilde{a}_{22}$		$\tilde{a}_{12}$	$\tilde{a}_{13}$	$\tilde{a}_{11}$		$\tilde{a}_{13}$	$\tilde{a}_{12}$	$\tilde{a}_{11}$

where $\tilde{a}_{ij}$ are the subsets that derive from the transposition of the corresponding a_{ij}'s of the original matrix and the proper replacement of their elements. Analogously the same holds for higher orders and there exist $n!$ isomorphisms for order n.

3 Examples and results

Let's assume that it must be verified whether or not the following two hypergroupoids are hypergroups.

$\cdot$	1	2	3
1	$\{1\}$	$\{2\}$	$\{1,2,3\}$
2	$\{2\}$	$\{1,2,3\}$	$\{1,2,3\}$
3	$\{1,2,3\}$	$\{1,2,3\}$	$\{1,3\}$

$\cdot$	1	2	3
1	$\{1,3\}$	$\{3\}$	$\{2\}$
2	$\{2\}$	$\{1,2,3\}$	$\{2\}$
3	$\{2\}$	$\{1\}$	$\{1,2,3\}$

we write:

```
In[1]:=<<HyperGroupTest.m
In[2]:=h1={{{1},{2},{1,2,3}},{{2},{1,2,3},{1,2,3}},
            {{1,2,3},{1,2,3},{1,3}}};
       h2={{{1,3},{3},{2}},{{2},{1,2,3},{2}},{{2},{1},{1,2,3}}};
In[3]:=HyperGroupTest[{h1,h2}]
Out[3]= {True,False}
```

From the last line (Out[3]) it derives that only the first hypercomposition defines a hypergroup in H. It is obvious though, that the second hypercomposition satisfies the reproductivity axiom and so the corresponding hypergroupoid is a quasi-hypergroup. We can verify this by using the `ReproductivityTest` function of the package in the appendix.

```
In[4]:=ReproductivityTest[h2]
Out[4]= True
```

The package in the appendix can handle hypergroupoids of arbitrary order. For $H = \{1, 2, 3, 4\}$ we check the following hypercomposition:

·	1	2	3	4
1	$\{1,3,4\}$	$\{1,3,4\}$	$\{2,3,4\}$	$\{1,2,3,4\}$
2	$\{1,2,4\}$	$\{1,2,3,4\}$	$\{1,2,4\}$	$\{1,3,4\}$
3	$\{1,2,4\}$	$\{1,2,3\}$	$\{3,4\}$	$\{1,2,4\}$
4	$\{2,3,4\}$	$\{2,3,4\}$	$\{1,2,3,4\}$	$\{1,2,3\}$

and we get

```
In[4]:=h3={{{1,3,4},{1,3,4},{2,3,4},{1,2,3,4}},
           {{1,2,4},{1,2,3,4},{1,2,4},{1,3,4}},
           {{1,2,4},{1,2,3},{3,4},{1,2,4}},
           {{2,3,4},{2,3,4},{1,2,3,4},{1,2,3}}}
In[5]:=HyperGroupTest[{h3}]
Out[5]= {True}
```

In order to evaluate N_3 all the 40-million hypergroupoids must be checked. First Λ is formed and then all the $7^3 = 343$ triads (in variable `a3`) and $7^6 = 343^2$ hexads (in variable `a6`) with elements from this set. Their combination in one and two rows respectively forms hypercomposition matrices of the form (1). Thus the memory requirements of a 7^9 length list containing the description of hypercompositions can be overcome. This can be done writing the lines below where variable `HyperGroups3` collects the Hypergroups we find.

```
In[6]:=lambda = Drop[Subsets[{1, 2, 3}], 1];
In[7]:=a6=Tuples[lambda, 6];
In[8]:=a3 = Tuples[lambda, 3];
In[9]:=HyperGroups3 = {};
In[10]:=Do[
          temp={Partition[Join[a3[[j1]], a6[[j2]]], 3]};
          If[HyperGroupTest[temp][[1]],HyperGroups3 = Join[HyperGroups3, temp]]
          , {j1, 1, 343}, {j2, 1, 343^2}];
In[11]:=Print[Length[HyperGroups3]]
Out[11]=23192
```

Interchanging in the above lines the function `HyperGroupTest` with the function `ReproductivityTest` found in the Appendix we counted 10323979 quasi-hypergroups of order 3. For order 2 we counted 35 quasi-hypergroups. Notice that for order 2 there are 14 hypergroups in 8 isomorphic classes.

A function that gives the $|H|!$ hypercompositions which form isomorphic hypergroupoids is given by accounting the observations in subsection (2.1) for every order.

```
IsomorphTest[a_List] :=
Module[{p, a1, len},
      len = Length[a]; p = Permutations[Range[1, len]];
      Return[Table[a1 = a;
                  a1 = ReplaceAll[a1, a1[[All, Table[j2, {j2, 1, len}]]] ->
```

```
                                a1[[All, p[[j1]]]]];
                    a1 = ReplaceAll[a1, a1[[Table[j2, {j2, 1, len}]]] ->
                                a1[[p[[j1]]]]];
                    a1 = ReplaceAll[a1, Flatten[Table[{p[[j1, j2]] -> j2}, {j2, 1, len}]]];
                    a1 = Table[Table[a1[[k1, k2]] = Sort[a1[[k1, k2]]], {k2, 1, len}],
                        {k1, 1, len}],
                {j1, 1, len!}]]
        ]
```

In order to count the number of the different non isomorphic classes of hypergroups of order 3, a 6 digit array, called `cardinalities` is used by the program. Each time the routine encounters a non isomorphic class, it drops it from `HyperGroups3`.

```
In[12]:=cardinalities = {0, 0, 0, 0, 0, 0};
In[13]:=While[Length[HyperGroups3] > 0,
            temp = Union[IsomorphTest[HyperGroups3[[1]]]];
            len = Length[Union[temp]];
            cardinalities[[len]] = cardinalities[[len]] + 1;
            HyperGroups3 = Complement[HyperGroups3, temp]
           ];
In[14]:=Total[cardinalities];
Out[14]=3999
In[15]:=Print[cardinalities];
Out[15]={6, 10, 244, 0, 0, 3739}
```

So we found that $S_3 = 3999$, and it is confirmed by the cardinalities of the isomorphic classes that $6 \cdot 1 + 10 \cdot 2 + 244 \cdot 3 + 3739 \cdot 6 = N_3$.

4 Conclusion

Generally speaking, few things are known about the construction of finite hypergroups. For example it is known that if $(H, \cdot)$ is a group or a hypergroup, then the $(H, \diamond)$ with $a \diamond b = a \cdot b \bigcup \{a, b\}$ is a hypergroup [2]. Thus using Cayley's theorem a family of finite hypergroups can be constructed based on finite groups. From the above analysis it derives that there are $7^9 = 40353607$ hypergroupoids of order 3, 23192 of these are hypergroups. The group of order 3 is among them, as well as the corresponding hypergroup constructed as above. The set of 23192 hypergroups is partitioned in 3999 equivalence classes. The 3739 of the above classes consists of 6 members, the 244 consists of 3 members, the 10 have 2 members and the last 6 are one member classes. The total hypergroup, that is the hypergroup in which the result of the hypercomposition consists always of all the elements of the hypergroup, is in the set C_1 of the six, one member, classes. In the same set belongs the B-hypergroup, i.e. the hypergroup in which the result of the hypercomposition consists only of the two elements which participate to the hypercomposition [3]. Relative to the B-hypergroup are two other non isomorphic hypergroups in which the hypercomposition is defined as follows (see [2])

$$ab = \{ \begin{matrix} \{a,b\} & \text{if } a \neq b \\ H & \text{if } a = b \end{matrix} \quad \text{and} \quad ab = \{ \begin{matrix} \{a,b\} & \text{if } a \neq b \\ H \setminus \{a\} & \text{if } a = b \end{matrix}$$

These hypergroups, when they have order 3, belong also to C_1. Generally it holds:

Proposition: *Let H be an arbitrary set with more than 2 elements. Then the hypercompositions*

$$ab = \{ \begin{matrix} H \text{ if a} \neq \text{b} \\ a \text{ if a} = \text{b} \end{matrix}, \; ab = \{ \begin{matrix} \{a,b\} & \text{if a} \neq \text{b} \\ H \setminus \{a\} & \text{if a} = \text{b} \end{matrix} \text{ and } ab = \{ \begin{matrix} \{a,b\} \text{ if a} \neq \text{b} \\ H \text{ if a} = \text{b} \end{matrix}$$

define in H three non isomorphic hypergroups.

It is worth mentioned that the number of the classes of hypergroups that can be constructed by the known Propositions and Theorems is very small comparing to the existing 3999 classes of hypergroups with three elements. Also one can notice that the ratio of hypergroups to hypergroupoids is exceptionally small since we meet on hypergroup in every 1740 hypergroupoids.

Appendix

The Mathematica package that implements the two basic properties (associativity and reproductivity) for testing if a hypergroupoid is indeed a Hypergroup follows.

```
BeginPackage["HyperGroupTest`"]; Clear["HyperGroupTest`*"];

HyperGroupTest::usage = "HyperGroupTest[LookUpTable] tests if
hypergroupoid operation given LookUpTable forms a HyperGroup"

Begin["`Private`"]; Clear["HyperGroupTest`Private`*"];

HyperGroupTest[LookUpTable_List] :=
Table[If[ReproductivityTest[LookUpTable[[j]]],
If[AssociativityTest[LookUpTable[[j]]], True,
                                       False],
       False], {j,1,Length[LookUpTable]}];

AssociativityTest[LookUpTable1_List] :=
    Module[{i,j,k,len,test},
           i = 1; j = 1; k = 1; test = True;len = Length[LookUpTable1];
           While[test && i<=len,
                      test = Union[Flatten[Union[Extract[LookUpTable1,
                                   Distribute[{LookUpTable1[[i, j]], {k}}, List]]]]] ==
                             Union[Flatten[Union[Extract[LookUpTable1,
                                   Distribute[{{i}, LookUpTable1[[j, k]]}, List]]]]];
                      k = k + 1;  If[k > len,
                                     k = 1; j = j + 1;
                                     If[j > len, i = i + 1; j = 1];
```

```
                              ];
                 ];
            Return[test]
           ];

ReproductivityTest[LookUpTable1_List] :=
      Union[Apply[Union,          LookUpTable1 , 1]] == {Range[1, Length[LookUpTable1]]} &&
      Union[Apply[Union,Transpose[LookUpTable1], 1]] == {Range[1, Length[LookUpTable1]]};

End[];
EndPackage[];
```

In the package above the function `AssociativityTest[ ]` is implemented by using `While`. In the most of the tested hypercompositions the property of associativity failed after the first 2 or 3 checks. Consequently it was not necessary to go through all 27 cases for hypergroups of order 3. Contrarily the function `ReproductivityTest[ ]` tested all rows and columns simultaneously according to property (2), since this does not increase computational time.

The program above can be used in order to construct hypergroups of any order. A random hypergroup of sixth order can be derived writing:

```
In[16]:=Do[temp =
  RandomChoice[
   Complement[Subsets[{1, 2, 3, 4, 5, 6}, {3, 6}], {{}}], {6, 6}];
 If[HyperGroupTest[{temp}][[1]], Print[temp]], {j1, 1, 30000000}]
```

And get the following hypergroup:

	1	2	3	4	5	6
1	$\{2,3,4\}$	$\{1,2,3,4,5,6\}$	$\{2,3,4,5,6\}$	$\{1,3,4,6\}$	$\{1,4,5,6\}$	$\{1,2,3,4,6\}$
2	$\{2,3,5,6\}$	$\{1,3,4,5\}$	$\{1,3,4\}$	$\{1,2,3,5,6\}$	$\{1,2,5\}$	$\{1,3,4,6\}$
3	$\{1,3,4,6\}$	$\{2,3,5,6\}$	$\{2,3,5,6\}$	$\{1,2,3,4\}$	$\{1,2,3,4,5,6\}$	$\{1,3,5,6\}$
4	$\{1,3,4,5,6\}$	$\{1,2,3,4,5,6\}$	$\{1,2,3\}$	$\{2,3,4,5\}$	$\{1,2,3,4\}$	$\{1,2,3,4,5,6\}$
5	$\{1,2,5,6\}$	$\{1,3,4,6\}$	$\{2,3,4,5\}$	$\{1,2,3,4,5,6\}$	$\{2,3,4,5,6\}$	$\{1,2,3,5\}$
6	$\{1,2,4,5,6\}$	$\{2,3,4\}$	$\{1,2,3,4,5,6\}$	$\{1,2,4\}$	$\{1,2,4,5,6\}$	$\{2,3,4,5\}$

References

[1] F. Marty, Sur in gènèralisation de la notion de group, Huitieme Congres des Matimaticiens scad, Stockholm 1934, pp. 45-59.

[2] C.G. Massouros, On the semi-subhypergroups of a hypergroup, Inter. J. Math. & Math. Sci., 14 (1991) pp. 293-304.

[3] G.G. Massouros, Automata and Hypermoduloids, Proceedings of the 5th Internat. Cong. on Algebraic Hyperstructures and Applications. pp. 251-266, Iasi 1993, Hadronic Press 1994.

[4] R. Migliorato, Ipergruppi di cardinalità 3 e isomorfismi di ipergruppoidi commutativi totalmente regolari, Atti Convegno su Ipergruppi, Udine 1985.

[5] J. Mittas, Hypergroupes Canoniques, Math, Balk. 2 (1972) pp. 165-179.

[6] G. Nordo, An algorithm on number of isomorphism classes of hypergroups of order 3, Italian J. Pure & Appl. Math., 2 (1997) pp. 37-42.

[7] S. Wolfram, The Mathematica Book, 5th ed., Wolfram Media, 2003.